

# Hyper

Oskar Wickström

fdfb323

## Contents

<b>1</b>	<b>Introduction</b>	<b>1</b>
1.1	Goals . . . . .	2
<b>2</b>	<b>Design</b>	<b>2</b>
2.1	Conn . . . . .	3
2.2	Middleware . . . . .	3
2.3	Response State Transitions . . . . .	4
2.4	Servers . . . . .	5
<b>3</b>	<b>Basics</b>	<b>6</b>
3.1	Request Body Reading . . . . .	6
3.2	Forms . . . . .	6
<b>4</b>	<b>Type-Level Routing</b>	<b>8</b>
4.1	A Single-Endpoint Example . . . . .	8
4.2	Routing Multiple Endpoints . . . . .	10
4.3	Content Negotiation . . . . .	12
4.4	Automatically Derived XHR Clients . . . . .	14
<b>5</b>	<b>NodeJS</b>	<b>14</b>
5.1	Monad Transformers . . . . .	14
<b>6</b>	<b>Testing</b>	<b>15</b>
<b>7</b>	<b>Contributing</b>	<b>16</b>

## 1 Introduction

Welcome to the Hyper documentation! Hyper is an experimental middleware architecture for HTTP servers written in PureScript. It is also a breeding ground for higher-level web server constructs, which tend to fall under the “framework”

category. Right now most things live in the main repository of Hyper, but they will most likely break free soon.

I hope you find something useful and inspiring in Hyper. If you want to get involved, see Contributing at the end of this document.

## 1.1 Goals

Composing middleware in NodeJS is a risky business. They mutate the HTTP request and response objects freely, and are often dependent on each others' side-effects. There are no guarantees that you have stacked the middleware functions in a sensible order, and it is often the case, in my experience, that misconfigured middleware takes a lot of time and effort to debug.

The goal of *Hyper* is to make use of row polymorphism and other tasty type system features in PureScript to enforce correctly stacked middleware in HTTP server applications. All effects of middleware should be reflected in the types to ensure that common mistakes cannot be made. A few examples could be:

- Incorrect ordering header and body writing
- Writing incomplete responses
- Overwriting headers
- Writing multiple responses
- Trying to consume a non-parsed request body
- Consuming a request body parsed as the wrong type
- Incorrect ordering of error handling middleware
- Incorrect ordering of middleware for sessions, authentication, authorization
- Missing authentication and/or authorization checks
- Linking, in an HTML anchor, to a resource that is not routed
- Posting, in an HTML form, to a resource that is not routed

Can we use the PureScript type system to eradicate this class of errors? Let's find out!

## 2 Design

We will start by looking at the central components of *Hyper*. While focusing heavily on safety, Hyper tries to provide an open API that can support multiple PureScript backends, and different styles of web applications.

The design of Hyper is inspired by a number of projects. The middleware chain lends much from *Plug*, an abstract HTTP interface in Elixir, that enables various HTTP libraries to inter-operate. You might also find similarities with *connect* in NodeJS. On the type system side, Hyper tries to bring in ideas from *Haskell*

and *Idris*, specifically the use of phantom types and GADTs to lift invariants to the type level and increase safety.

## 2.1 Conn

A *Conn*, short for “connection”, models the entirety of a connection between the HTTP server and the user agent, both request and response.

```
type Conn req res components =  
  { request :: req  
    , response :: res  
    , components :: components  
  }
```

The `request` and `response` hold the values representing the HTTP request and response, respectively. The purpose of the `components` field, however, is not that obvious. It is used for things not directly related to the HTTP, but nonetheless related to the act of responding to the HTTP request. A middleware can add information the `Conn` using components, like providing authentication or authorization values. The types of these components then becomes part of the `Conn` type, and you get compile-time guarantees when using the provided components.

## 2.2 Middleware

A *middleware* is an *indexed monadic action* transforming one `Conn` to another `Conn`. It operates in some base monad `m`, and is indexed by `i` and `o`, the *input* and *output* `Conn` types of the middleware action.

```
newtype Middleware m i o a = ...
```

The input and output type parameters are used to ensure that a `Conn` is transformed, and that side-effects are performed, correctly, throughout the middleware chain.

Being able to parameterize `Middleware` with some type `m`, you can customize the chain depending on the needs of your middleware and handlers. Applications can use monad transformers to track state, provide configuration, gather metrics, and much more, in the chain of middleware.

Middleware are composed using `ibind`, the indexed monadic version of `bind`. The simplest way of composing middleware is by chaining them with `:*>`, from `Control.IxMonad`. See [purescript-indexed-monad](https://pursuit.purescript.org/packages/purescript-indexed-monad/0.1.1)<sup>1</sup> for more information.

---

<sup>1</sup><https://pursuit.purescript.org/packages/purescript-indexed-monad/0.1.1>

```

writeStatus statusOK
:*> closeHeaders
:*> respond "We're composing middleware!"

```

If you want to feed the return value of one middleware into another, use `:>=>`, the infix operator alias for `mbind`.

```

getUser :>=> renderUser

```

You can also rebind the *do block* syntax to use `ibind` instead of regular `bind`.

```

do
  user <- getUser
  writeStatus statusOK
  closeHeaders
  respond ("User: " <> user.name)
where bind = ibind

```

## 2.3 Response State Transitions

The `writer` field in the `response` record of a `Conn` is a value provided by the server backend. Middleware often constrain the `writer` field to be a value implementing the `ResponseWriter` type class. This makes it possible to provide response writing abstractions without depending on a specific server backend.

The state of a response writer is tracked in its type parameter. This state tracking, and the type-indexed middleware using the response writer, guarantee correctness in response handling, preventing incorrect ordering of headers and body writes, incomplete responses, or other such mistakes. Let us have a look at the type signatures of some of response writing functions in `Hyper.Response`.

We see that `headers` takes a traversable collection of headers, and gives back a middleware that, given a connection where headers are ready to be written (`HeadersOpen`), writes all specified headers, writes the separating CRLF before the HTTP body, and marks the state of the response writer as being ready to write the body (`BodyOpen`).

```

headers :: forall t m req res rw c.
  (Traversable t, Monad m, ResponseWriter rw m b) =>
  t Header
-> Middleware
  m
  (Conn req { writer :: rw HeadersOpen | res } c)
  (Conn req { writer :: rw BodyOpen | res } c)

```

## Unit

To be used in combination with `headers`, the `respond` function takes some `Response m r b`, and gives back a middleware that, given a connection *where all headers have been written*, writes a response, and *marks the state of the response writer as ended*.

```
respond :: forall m r b req res rw c.
  (Monad m, Response m r b, ResponseWriter rw m b) =>
  r
-> Middleware
  m
  (Conn req { writer :: rw BodyOpen | res } c)
  (Conn req { writer :: rw ResponseEnded | res } c)
  Unit
```

The `Response` type class describes types that can be written as responses. It takes three type parameters, where `b` is the target type, `m` is a base monad for the `Middleware` returned, and `r` is the original response type,

```
class Response b m r where
  toResponse :: forall i. r -> Middleware m i b
```

This mechanism allows servers to provide specific types for the response body, along with instances for common response types. When using the Node server, which has a response body type wrapping `Buffer`, you can still respond with a `String` or `HTML` value directly.

Aside from convenience in having a single function for most response types and servers, the polymorphism of `respond` lets middleware be decoupled from specific servers. It only requires an instance matching the response type used by the middleware and the type required by the server.

## 2.4 Servers

Although `Hyper` middleware can be applied directly to `Conn` values using `runMiddleware`, you likely want a *server* to run your middleware. `Hyper` tries to be as open as possible when it comes to servers – your application, and the middleware it depends on, should not be tied to a specific server. This allows for greater reuse and the ability to test entire applications without running the “real” server. Currently `Hyper` bundles a NodeJS server, as well as a test server.

## 3 Basics

This chapter walks through some of the basic and more lower-level features of Hyper. We look at how we can use the `Conn` and `Middleware` types to gain type safety “in the small”, and how to implement common things needed in web servers.

### 3.1 Request Body Reading

The `RequestBodyReader` type class has one operation, `readBody`, which supports different servers to provide different types of request body values.

```
class RequestBodyReader r m b | r -> b where
  readBody
    :: forall req res c.
       Middleware
         m
         (Conn { body :: r | req } res c)
         (Conn { body :: r | req } res c)
         b
```

Given that there is an instance for the body `b`, and the return type `r`, we can use this middleware together with other middleware, like so:

```
onPost =
  readBody :>=>
  case _ of
    "" ->
      writeStatus statusBadRequest
      :*> closeHeaders
      :*> respond "... anyone there?"
    msg ->
      writeStatus statusBadRequest
      :*> closeHeaders
      :*> respond ("You said: " <> msg)
```

### 3.2 Forms

When working with form data, we often want to serialize and deserialize forms as custom data types, instead of working with the key-value pairs directly. The `ToForm` and `FromForm` type classes abstracts serialization and deserialization to form data, respectively.

We first declare our data types, and some instance which we will need later.

```
data MealType = Vegan | Vegetarian | Omnivore | Carnivore

derive instance genericMealType :: Generic MealType _
instance eqMealType :: Eq MealType where eq = genericEq
instance showMealType :: Show MealType where show = genericShow

newtype Order = Order { beers :: Int, meal :: MealType }
```

In this example we will only deserialize forms, and thus we only need the FromForm instance.

```
instance fromFormOrder :: FromForm Order where
  fromForm form = do
    beers <- required "beers" form >>= parseBeers
    meal <- required "meal" form >>= parseMealType
    pure (Order { beers: beers, meal: meal })
  where
    parseBeers s =
      maybe
        (throwError ("Invalid number: " <> s))
        pure
        (Int.fromString s)

    parseMealType =
      case _ of
        "Vegan" -> pure Vegan
        "Vegetarian" -> pure Vegetarian
        "Omnivore" -> pure Omnivore
        "Carnivore" -> pure Carnivore
        s -> throwError ("Invalid meal type: " <> s)
```

Now we are ready to write our handler. We use `parseFromForm` to get a value of type `Either String Order`, where the `String` explains parsing errors. By pattern matching using record field puns, we extract the `beers` and `meal` values, and respond based on those values.

```
onPost =
  parseFromForm :>>=
  case _ of
    Left err ->
      writeStatus statusBadRequest
      :*> closeHeaders
      :*> respond (err <> "\n")
```

```

Right (Order { beers, meal })
| meal == Omnivore || meal == Carnivore ->
  writeStatus statusBadRequest
  :*> closeHeaders
  :*> respond "Sorry, we do not serve meat here.\n"
| otherwise ->
  writeStatus statusBadRequest
  :*> closeHeaders
  :*> respond ("One " <> show meal <> " meal and "
               <> show beers <> " beers coming up!\n")

```

Let's try this server out at the command line.

```

$ curl -X POST -d 'beers=6' http://localhost:3000
Missing field: meal
$ curl -X POST -d 'meal=Omnivore&beers=6' http://localhost:3000
Sorry, we do not serve meat here.
$ curl -X POST -d 'meal=Vegetarian&beers=6' http://localhost:3000
One Vegetarian meal and 6 beers coming up!

```

## 4 Type-Level Routing

`Hyper.Routing` provides an API for expressing web application routing as types, much like `Servant`<sup>2</sup> does. By using routing types you get static guarantees about having handled all cases. You also get a lot of stuff for free, such as type-safe parameters for handlers, and automatically generated type-safe URIs to endpoints.

### 4.1 A Single-Endpoint Example

Let's say we want to render a home page as HTML. We start out by declaring the endpoint data type `Home`, and the structure of our site:

```

data Home = Home

type Site1 = Get HTML Home

```

`Get HTML Home` is a routing type with only one endpoint, rendering a `Home` value as HTML. So where does the `Home` value come from? We provide it using a *handler*. A handler for `Site1` would be some value of the following type:

---

<sup>2</sup><https://haskell-servant.github.io>



```
forall m. Monad m => ExceptT RoutingError m Home
```

We can construct such a value using `pure` and a `Home` value:

```
home :: forall m. Applicative m => m Home
home = pure Home
```

Nice! But what comes out on the other end? We need something that renders the `Home` value as HTML. By providing an instance of `EncodeHTML` for `Home`, we instruct the endpoint how to render.

```
instance encodeHTMLHome :: EncodeHTML Home where
  encodeHTML Home =
    p (text "Welcome to my site!")
```

The `HTML` type is a phantom type, only used as a marker type, and the actual markup is written in the `MarkupM` DSL from `purescript-smolder`<sup>3</sup>.

We are getting ready to create the server. First, we need a value-level representation of the `Site1` type, to be able to pass it to the `router` function. For that we use `Proxy`<sup>4</sup>. Its documentation describes it as follows:

The `Proxy` type and values are for situations where type information is required for an input to determine the type of an output, but where it is not possible or convenient to provide a value for the input.

We create a top-level definition of the type `Proxy Site1` with the value constructor `Proxy`.

```
site1 :: Proxy Site1
site1 = Proxy
```

We pass the proxy, our handler, and the `onRoutingError` function for cases where no route matched the request, to the `router` function.

```
onRoutingError status msg =
  writeStatus status
  :*> contentType textHTML
  :*> closeHeaders
  :*> respond (maybe "" id msg)

siteRouter = router site1 home onRoutingError
```

---

<sup>3</sup><https://github.com/bodil/purescript-smolder>

<sup>4</sup><https://pursuit.purescript.org/packages/purescript-proxy/1.0.0/docs/Type.Proxy>

The value returned by `router` is regular middleware, ready to be passed to a server.

```
main :: forall e. Eff (http :: HTTP, console :: CONSOLE, buffer
  ↳ :: BUFFER | e) Unit
main =
  runServer defaultOptions {} siteRouter
  where
```

## 4.2 Routing Multiple Endpoints

Real-world servers often need more than one endpoint. Let's define a router for an application that shows a home page with links, a page listing users, and a page rendering a specific user.

```
data Home = Home

data AllUsers = AllUsers (Array User)

newtype User = User { id :: Int, name :: String }

type Site2 =
  Get HTML Home
  :<|> "users" :/ Get HTML AllUsers
  :<|> "users" :/ Capture "user-id" Int :> Get HTML User

site2 :: Proxy Site2
site2 = Proxy
```

Let's go through the new constructs used:

- `:<|>` is a type operator that separates *alternatives*. A router for this type will try each route in order until one matches.
- `:/` separates a literal path segment and the rest of the endpoint type.
- `Capture` takes a descriptive string and a type. It takes the next available path segment and tries to convert it to the given type. Each capture in an endpoint type corresponds to an argument in the handler function.
- `:>` separates an endpoint modifier, like `Capture`, and the rest of the endpoint type.

We define handlers for our routes as regular functions on the specified data types, returning `ExceptT RoutingError m a` values, where `m` is the monad of our middleware, and `a` is the type to render for the endpoint.

```
home :: forall m. Monad m => ExceptT RoutingError m Home
```

```

home = pure Home

allUsers :: forall m. Monad m => ExceptT RoutingError m AllUsers
allUsers = AllUsers <$> getUsers

getUser :: forall m. Monad m => Int -> ExceptT RoutingError m
  ↳ User
getUser id' =
  find userWithId <$> getUsers >=
  case _ of
    Just user -> pure user
    Nothing ->
      throwError (HTTPError { status: statusNotFound
                              , message: Just "User not found."
                              })

  where
    userWithId (User u) = u.id == id'

```

As in the single-endpoint example, we want to render as HTML. Let's create instances for our data types. Notice how we can create links between routes in a type-safe manner.

```

instance encodeHTMLHome :: EncodeHTML Home where
  encodeHTML Home =
    case linksTo site2 of
      _ :<|> allUsers' :<|> _ ->
        p do
          text "Welcome to my site! Go check out my "
          linkTo allUsers' (text "Users")
          text "."

instance encodeHTMLAllUsers :: EncodeHTML AllUsers where
  encodeHTML (AllUsers users) =
    div do
      h1 (text "Users")
      ul (traverse_ linkToUser users)
    where
      linkToUser (User u) =
        case linksTo site2 of
          _ :<|> _ :<|> getUser' ->
            li (linkTo (getUser' u.id) (text u.name))

instance encodeHTMLUser :: EncodeHTML User where
  encodeHTML (User { name }) =
    h1 (text name)

```

The pattern match on the value returned by `linksTo` must match the structure of the routing type. We use `:<|>` to pattern match on links. Each matched link will have a type based on the corresponding endpoint. `getUser` in the previous code has type `Int -> URI`, while `allUsers` has no captures and thus has type `URI`.

We are still missing `getUsers`, our source of `User` values. In a real application it would probably be a database query, but for this example we simply hard-code some famous users of proper instruments.

```
getUsers :: forall m. Applicative m => m (Array User)
getUsers =
  pure
    [ User { id: 1, name: "John Paul Jones" }
    , User { id: 2, name: "Tal Wilkenfeld" }
    , User { id: 3, name: "John Patitucci" }
    , User { id: 4, name: "Jaco Pastorious" }
    ]
```

Almost done! We just need to create the router, and start a server.

```
main :: forall e. Eff (http :: HTTP, console :: CONSOLE, buffer
  ⇨ :: BUFFER | e) Unit
main =
  let otherSiteRouter =
      router site2 (home :<|> allUsers :<|> getUser)
      ⇨ onRoutingError

      onRoutingError status msg =
        writeStatus status
        :*> contentType textHTML
        :*> closeHeaders
        :*> respond (maybe "" id msg)

  in runServer defaultOptions {} otherSiteRouter
```

Notice how the composition of handler functions, using the value-level operator `:<|>`, matches the structure of our routing type. If we fail to match the type we get a compile error.

### 4.3 Content Negotiation

By specifying alternative content types for an endpoint, Hyper can choose a response and content type based on the request `Accept` header. This is called *content negotiation*. Instead of specifying a single type, like `HTML` or `JSON`, we

provide alternatives using `:<|>`. All content types must have `MimeRender` instances for the response body type.

```
type Site3 =
  Get HTML Home
  :<|> "users" :/ Get (HTML :<|> JSON) AllUsers
  :<|> "users" :/ Capture "user-id" Int :> Get (HTML :<|> JSON)
  ↪ User
```

By making requests to this site, using `Accept` headers, we can see how the router chooses the matching content type (output formatted and shortened for readability).

```
$ curl -H 'Accept: application/json' http://localhost:3000/users
[
  {
    "name": "John Paul Jones",
    "id": "1"
  },
  {
    "name": "Tal Wilkenfeld",
    "id": "2"
  },
  ...
]
```

There's experimental support for *wildcards* and *qualities* as well.

```
$ curl -H 'Accept: text/*;q=1.0' http://localhost:3000/users
<div>
  <h1>Users</h1>
  <ul>
    <li><a href="/users/1">John Paul Jones</a></li>
    <li><a href="/users/2">Tal Wilkenfeld</a></li>
    ...
  </ul>
</div>
```

## 4.4 Automatically Derived XHR Clients

As we represent routing as types, we can derive XHR clients from those types. In a client-side application, for example one written using Pux<sup>5</sup> or Halogen<sup>6</sup>, you can use the `purescript-hyper-routing-xhr`<sup>7</sup> library to derive such client functions. You get functions with type-safe parameters and encoding/decoding, using your routed types, and results in the `Aff`<sup>8</sup> monad, which are easily integrated into most frameworks.

## 5 NodeJS

The server in `Hyper.Node.Server` wraps the `http` module in NodeJS, and serves middleware using the `Aff` monad. Here is how you can start a Node server:

```
let
  app =
    writeStatus (Tuple 200 "OK")
    :*> closeHeaders
    :*> respond "Hello there!"
in runServer defaultOptions {} app
```

As seen above, `runServer` takes a record of options, an initial *components* record, and your application middleware. If you want to do logging on server startup, and on any request handling errors, use `defaultOptionsWithLogging`.

### 5.1 Monad Transformers

You might want to use a monad transformer stack in your application, for instance as a way to pass configuration, or to accumulate some state in the chain of middleware. The underlying monad of `Middleware` is parameterized for this exact purpose. When running the NodeJS server with monad transformers, you need to pass a function in the options record, in the `runM` field.

The following code runs a middleware using the `ReaderT` monad transformer. Note that the `runAppM` function might need to be defined at the top-level to please the type checker.

```
type MyConfig = { thingToSay String }
```

```
runAppM
```

---

<sup>5</sup><https://www.purescript-pux.org>

<sup>6</sup><https://github.com/slamdata/purescript-halogen>

<sup>7</sup><https://github.com/owickstrom/purescript-hyper-routing-xhr>

<sup>8</sup><https://github.com/slamdata/purescript-aff>

```

    e a.
    ReaderT MyConfig (Aff e) a→
    Aff e a
runAppM = flip runReaderT { thingToSay: "Hello, ReaderT!" }

main :: forall e. Eff (console :: CONSOLE, http :: HTTP | e) Unit
main =
    let app = do
        config ← lift' ask
        writeStatus statusOK
        closeHeaders
        respond config.thingToSay
        where bind = ibind

    options = defaultOptionsWithLogging { runM = runAppM }

    in runServer options {} app

```

In a real-world application the configuration type `MyConfig` could hold a database connection pool, or settings read from the environment, for example.

## 6 Testing

When running tests you might not want to start a full HTTP server and send requests using an HTTP client. Instead you can use the server in `Hyper.Test.TestServer`. It runs your middleware directly on `Conn` values, and collects the response using a `Writer` monad. You get back a `TestResponse` from which you can extract the status code, headers, and the response body.

```

it "responds with a friendly message" do
    conn <- { request: {}
            , response: { writer: testResponseWriter }
            , components: {}
            }
    # evalMiddleware app
    # testServer
    testStatus conn `shouldEqual` Just statusOK
    testStringBody conn `shouldEqual` "Hello there!"

```

## 7 Contributing

While Hyper is currently an experiment, and in constant flux, you are welcome to contribute. Please post ideas and start discussions using the issue tracker on GitHub<sup>9</sup>. You can also contact Oskar Wickström<sup>10</sup> directly for design discussions. If this project grows, we can setup a mailing list, or other some other means of communication.

Please note that sending pull requests without first discussing the design is probably a waste of time, if not only fixing simple things like typos.

---

<sup>9</sup><https://github.com/owickstrom/hyper/issues>

<sup>10</sup><https://wickstrom.tech/about.html>