

# Hyper

Oskar Wickström

## Contents

<b>1</b>	<b>Goals</b>	<b>1</b>
<b>2</b>	<b>Design</b>	<b>2</b>
2.1	Conn . . . . .	2
2.2	Middleware . . . . .	2
2.3	Response State Transitions . . . . .	3
<b>3</b>	<b>Request Body Parsing</b>	<b>4</b>
<b>4</b>	<b>Type-Level Routing</b>	<b>5</b>
4.1	A Single-Endpoint Example . . . . .	5
4.2	Routing Multiple Endpoints . . . . .	7
4.3	Content Negotiation . . . . .	10
<b>5</b>	<b>Servers</b>	<b>11</b>
5.1	NodeJS . . . . .	11
5.2	Testing . . . . .	11
<b>6</b>	<b>Contributing</b>	<b>12</b>

## 1 Goals

Composing middleware in NodeJS is a risky business. They mutate the HTTP request and response objects freely, and are often dependent on each others' side-effects. There are no guarantees that you have stacked the middleware functions in a sensible order, and it is often the case, in my experience, that misconfigured middleware takes a lot of time and effort to debug.

The goal of *Hyper* is to make use of row polymorphism and other tasty type system features in PureScript to enforce correctly stacked middleware in HTTP server applications. All effects of middleware should be reflected in the types to ensure that common mistakes cannot be made. A few examples could be:

- Incorrect ordering header and body writing

- Writing incomplete responses
- Overwriting headers
- Writing multiple responses
- Trying to consume a non-parsed request body
- Consuming a request body parsed as the wrong type
- Incorrect ordering of error handling middleware
- Incorrect ordering of middleware for sessions, authentication, authorization
- Missing authentication and/or authorization checks
- Linking, in an HTML anchor, to a resource that is not routed
- Posting, in an HTML form, to a resource that is not routed

Can we use the PureScript type system to eradicate this class of errors? Let's find out!

## 2 Design

We will start by looking at the central components of *Hyper*. While focusing heavily on safety, Hyper tries to provide an open API that can support multiple PureScript backends, and different styles of web applications.

The design of Hyper is inspired by a number of projects. The middleware chain lends much from *Plug*, an abstract HTTP interface in Elixir, that enables various HTTP libraries to inter-operate. You might also find similarities with *connect* in NodeJS. On the type system side, Hyper tries to bring in ideas from *Haskell* and *Idris*, specifically the use of phantom types and GADTs to lift invariants to the type level and increase safety.

### 2.1 Conn

A *Conn*, short for “connection”, models the entirety of a connection between the HTTP server and the user agent, both request and response.

```
type Conn req res components = { request :: req
                                , response :: res
                                , components :: components
                                }
```

### 2.2 Middleware

A *middleware* is a function transforming a *Conn* to another *Conn*, in some monadic type *m*. The *Middleware* type synonym encapsulates this concept, but it is still a regular function.

```
type Middleware m c c' = c -> m c'
```

Being able to parameterize `Conn` with some type `m`, you can customize the chain depending on the needs of your middleware and handlers. Applications can use monad transformers to track state, provide configuration, gather metrics, and much more, in the chain of middleware.

## 2.3 Response State Transitions

The `writer` field in the `response` record of a `Conn` is a value provided by the server backend. Functions usually constrain the `writer` field to be a value implementing the `ResponseWriter` type class. This makes it possible to provide response writing abstractions without depending on a specific server backend.

The state of a response writer is tracked in its type parameter. This state tracking, and the type signatures of functions using the response writer, guarantee correctness in response handling, preventing incorrect ordering of headers and body writes, incomplete responses, or other such mistakes. Let us have a look at the type signatures of some of response writing functions in `Hyper.Response`.

We see that `headers` takes a traversable collection of headers, and gives back a middleware that, given a connection *where headers are ready to be written*, writes all specified headers, writes the separating CRLF before the HTTP body, and *marks the state of the response writer as headers being closed*.

```
headers :: forall t m req res rw c.
  (Traversable t, Monad m, ResponseWriter rw m b) =>
  t Header
-> Middleware
  m
  (Conn req { writer :: rw HeadersOpen | res } c)
  (Conn req { writer :: rw BodyOpen | res } c)
```

To be used in combination with `headers`, the `respond` function takes some `Response m r b`, and gives back a middleware that, given a connection *where all headers have been written*, writes a response, and *marks the state of the response writer as ended*.

```
respond :: forall m r b req res rw c.
  (Monad m, Response m r b, ResponseWriter rw m b) =>
  r
-> Middleware
  m
  (Conn req { writer :: rw BodyOpen | res } c)
  (Conn req { writer :: rw ResponseEnded | res } c)
```

The `Response` type class describes types that can be written as responses. It takes three type parameters, where `r` is the original response type, `m` is usually an `Applicative` or a `Monad` in which the transformation can be performed, and `b` is the target type.

```
class Response m r b | r -> b where
  toResponse :: r -> m b
```

This mechanism allows servers to provide specific types for the response body, along with instances for common response types. When using the Node server, which has a response body type wrapping `Buffer`, you can still respond with a `String` or `HTML` value directly.

Aside from convenience in having a single function for most response types and servers, the polymorphism of `respond` lets middleware be decoupled from specific servers. It only requires an instance matching the response type used by the middleware and the type required by the server.

### 3 Request Body Parsing

The request body is, when using the Node server, initially a `RequestBody` in the connection. The user explicitly chooses to read and parse the body with a given parser, which returns a new connection of a type reflecting the action. The following type signature resides in `Hyper.Node.Server`, and shows how a request body can be read into a `String`. The `Aff` monad, and the `AVAR` effect, is used to accomplish this asynchronously in the case of the Node server.

```
readBodyAsString
  :: forall e req res c.
  Middleware
  (Aff (http :: HTTP, err :: EXCEPTION, avar :: AVAR | e))
  (Conn { body :: RequestBody
        , contentLength :: Maybe Int
        | req
        } res c)
  (Conn { body :: String
        , contentLength :: Maybe Int | req
        } res c)
```

A simple form parser can use `readBodyAsString` to convert the body a more useful format for the application. The following function checks the `Content-Type` header in the request, splits the request body, builds up a `Form` value, and finally using that value for the `body` field in the resulting `Conn`. The form body has type `Either Error Form` to represent invalid forms.

```

parseForm :: forall m req res c.
  Applicative m =>
  Middleware
  m
  (Conn { body   String
        , headers :: StrMap String
        | req
        } res c)
  (Conn { body   Either Error Form
        , headers :: StrMap String
        | req
        }
    res
    c)
parseForm conn = ...

```

More efficient parsers, directly operating on the `RequestBody`, instead of `String`, can of course be built as well.

## 4 Type-Level Routing

`Hyper.Routing` provides an API for expressing web application routing as types, much like `Servant`<sup>1</sup> does. By using routing types you get static guarantees about having handled all cases. You also get a lot of stuff for free, such as type-safe parameters for handlers, and automatically generated type-safe URIs to endpoints.

*Future work might include automatic API documentation and sitemaps, automatic deriving of PureScript AJAX clients, and more.*

### 4.1 A Single-Endpoint Example

Let's say we want to render a home page as HTML. We start out by declaring the endpoint data type `Home`, and the structure of our site:

```

data Home = Home

type Site1 = Get HTML Home

```

`Get HTML Home` is a routing type with only one endpoint, rendering a `Home` value as HTML. So where does the `Home` value come from? We provide it using a *handler*. A handler for `Site1` would be some value of the following type:

---

<sup>1</sup><https://haskell-servant.github.io>

```
forall m. Monad m => ExceptT RoutingError m Home
```

We can construct such a value using `pure` and a `Home` value:

```
home = pure Home
```

Nice! But what comes out on the other end? We need something that renders the `Home` value as HTML. By providing an instance of `EncodeHTML` for `Home`, we instruct the endpoint how to render.

```
instance encodeHTMLHome :: EncodeHTML Home where
  encodeHTML Home =
    p [] [ text "Welcome to my site!" ]
```

We are getting ready to create the server. First, we need a value-level representation of the `Site1` type, to be able to pass it to the `router` function. For that we use `Proxy`<sup>2</sup>. Its documentation describes it as follows:

The `Proxy` type and values are for situations where type information is required for an input to determine the type of an output, but where it is not possible or convenient to provide a value for the input.

We create a top-level definition of the type `Proxy Site1` with the value constructor `Proxy`.

```
site1 :: Proxy Site1
site1 = Proxy
```

We pass the proxy, our handler, and the `onRoutingError` function for cases where no route matched the request, to the `router` function.

```
onRoutingError status msg =
  writeStatus status
  >=> contentType textHTML
  >=> closeHeaders
  >=> respond (maybe "" id msg)

site1Router = router site1 home onRoutingError
```

The value returned by `router` is regular middleware, ready to be passed to a server.

```
main =
```

---

<sup>2</sup><https://pursuit.purescript.org/packages/purescript-proxy/1.0.0/docs/Type.Proxy>

```

runServer defaultOptions onListening onRequestError {}
  ↪ site1Router
where
  onListening (Port port) =
    log ("Listening on http://localhost:" <> show port)

  onRequestError err =
    log ("Request failed: " <> show err)

```

## 4.2 Routing Multiple Endpoints

Real-world servers often need more than one endpoint. Let's define a router for an application that shows a home page with links, a page listing users, and a page rendering a specific user.

```

data Home = Home

data AllUsers = AllUsers (Array User)

newtype User = User { id :: Int, name :: String }

type Site2 =
  Get HTML Home
  :<|> "users" :/ Get HTML AllUsers
  :<|> "users" :/ Capture "user-id" Int :> Get HTML User

site2 :: Proxy Site2
site2 = Proxy

```

Let's go through the new constructs used:

- `:<|>` is a type operator that separates *alternatives*. A router for this type will try each route in order until one matches.
- `:/` separates a literal path segment and the rest of the endpoint type.
- `Capture` takes a descriptive string and a type. It takes the next available path segment and tries to convert it to the given type. Each capture in an endpoint type corresponds to an argument in the handler function.
- `:>` separates an endpoint modifier, like `Capture`, and the rest of the endpoint type.

We define handlers for our routes as regular functions on the specified data types, returning `ExceptT RoutingError m a` values, where `m` is the monad of our middleware, and `a` is the type to render for the endpoint.

```

home :: forall m. Monad m => ExceptT RoutingError m Home

```

```

home = pure Home

allUsers :: forall m. Monad m => ExceptT RoutingError m AllUsers
allUsers = AllUsers <$> getUsers

getUser :: forall m. Monad m => Int -> ExceptT RoutingError m
  ↳ User
getUser id' =
  find userWithId <$> getUsers >=
  case _ of
    Just user -> pure user
    Nothing ->
      throwError (HTTPError { status: statusNotFound
                              , message: Just "User not found."
                              })

where
  userWithId (User u) = u.id == id'

```

As in the single-endpoint example, we want to render as HTML. Let's create instances for our data types. Notice how we can create links between routes in a type-safe manner.

```

instance encodeHTMLHome :: EncodeHTML Home where
  encodeHTML Home =
    case linksTo site2 of
      _ :<|> allUsers' :<|> _ ->
        p [] [ text "Welcome to my site! Go check out my "
                , linkTo allUsers' [ text "Users" ]
                , text "."
              ]

instance encodeHTMLAllUsers :: EncodeHTML AllUsers where
  encodeHTML (AllUsers users) =
    element_ "div" [ h1 [] [ text "Users" ]
                    , ul [] (map linkToUser users)
                  ]

  where
    linkToUser (User u) =
      case linksTo site2 of
        _ :<|> _ :<|> getUser' ->
          li [] [ linkTo (getUser' u.id) [ text u.name ] ]

instance encodeHTMLUser :: EncodeHTML User where
  encodeHTML (User { name }) =
    h1 [] [ text name ]

```



The pattern match on the value returned by `linksTo` must match the structure of the routing type. We use `:<|>` to pattern match on links. Each matched link will have a type based on the corresponding endpoint. `getUser` in the previous code has type `Int -> URI`, while `allUsers` has no captures and thus has type `URI`.

We are still missing `getUsers`, our source of `User` values. In a real application it would probably be a database query, but for this example we simply hard-code some famous users of proper instruments.

```
getUsers :: forall m. Applicative m => m (Array User)
getUsers =
  pure
  [ User { id: 1, name: "John Paul Jones" }
  , User { id: 2, name: "Tal Wilkenfeld" }
  , User { id: 3, name: "John Patitucci" }
  , User { id: 4, name: "Jaco Pastorious" }
  ]
```

Almost done! We just need to create the router, and start a server.

```
main =
  let site2Router =
      router site2 (home :<|> allUsers :<|> getUser)
      ↪ onRoutingError

      onRoutingError status msg =
        writeStatus status
        >=> contentType textHTML
        >=> closeHeaders
        >=> respond (maybe "" id msg)

      onListening (Port port) =
        log ("Listening on http://localhost:" <> show port)

      onRequestError err =
        log ("Request failed: " <> show err)

  in runServer defaultOptions onListening onRequestError {}
  ↪ site2Router
```

Notice how the composition of handler functions, using the value-level operator `:<|>`, matches the structure of our routing type. If we fail to match the type we get a compile error.

### 4.3 Content Negotiation

By specifying alternative content types for an endpoint, Hyper can choose a response and content type based on the request `Accept` header. This is called *content negotiation*. Instead of specifying a single type, like `HTML` or `Json`, we provide alternatives using `:<|>`. All content types must have `MimeRender` instances for the response body type.

```
type Site3 =
  Get (HTML :<|> Json) Home
  :<|> "users" :/ Get (HTML :<|> Json) AllUsers
  :<|> "users" :/ Capture "user-id" Int :> Get (HTML :<|> JSON)
  ⇨ User
```

By making requests to this site, using `Accept` headers, we can see how the router chooses the matching content type (output formatted and shortened for readability).

```
$ curl -H 'Accept: application/json' http://localhost:3000/users
[
  {
    "name": "John Paul Jones",
    "id": "1"
  },
  {
    "name": "Tal Wilkenfeld",
    "id": "2"
  },
  ...
]
```

There's experimental support for *wildcards* and *qualities* as well.

```
$ curl -H 'Accept: text/*;q=1.0' http://localhost:3000/users
<div>
  <h1>Users</h1>
  <ul>
    <li><a href="/users/1">John Paul Jones</a></li>
    <li><a href="/users/2">Tal Wilkenfeld</a></li>
    ...
  </ul>
</div>
```

## 5 Servers

Although Hyper middleware are regular functions, which can be applied to `Conn` values, you often want a *server* to run your middleware. Hyper tries to be as open as possible when it comes to servers – your application, and the middleware it depends on, should not be tied to a specific server. This allows for greater reuse and the ability to test entire applications without running the “real” server.

### 5.1 NodeJS

The server in `Hyper.Node.Server` wraps the `http` module in NodeJS, and serves middleware using the `Aff` monad. Here is how you can start a Node server:

```
let
  onListening (Port port) =
    log ("Listening on http://localhost:" <> show port)
  onRequestError err =
    log ("Request failed: " <> show err)
  app =
    writeStatus (Tuple 200 "OK")
    >=> closeHeaders
    >=> respond "Hello there!"
in runServer defaultOptions onListening onRequestError {} app
```

As seen above, `runServer` takes a record of options, two callbacks, an initial *components* record, and your application middleware.

### 5.2 Testing

When running tests you might not want to start a full HTTP server and send requests using an HTTP client. Instead you can use the server in `Hyper.Test.TestServer`. It runs your middleware directly on `Conn` values, and collects the response using a `Writer` monad. You get back a `TestResponse` from which you can extract the status code, headers, and the response body.

```
it "responds with a friendly message" do
  conn <- { request: {}
    , response: { writer: testResponseWriter }
    , components: {}
  }
  # app
  # testServer
  testStatus conn `shouldEqual` Just statusOK
  testStringBody conn `shouldEqual` "Hello there!"
```

## 6 Contributing

While Hyper is currently an experiment, and in constant flux, you are welcome to contribute. Please post ideas and start discussions using the issue tracker on GitHub<sup>3</sup>. You can also contact Oskar Wickström<sup>4</sup> directly for design discussions. If this project grows, we can setup a mailing list, or other some other means of communication.

Please note that sending pull requests without first discussing the design is probably a waste of time, if not only fixing simple things like typos.

---

<sup>3</sup><https://github.com/owickstrom/hyper/issues>

<sup>4</sup><https://wickstrom.tech/about.html>